



**HAL**  
open science

# A Simpler and Exact Sampling Routine for the GGX Distribution of Visible Normals

Eric Heitz

► **To cite this version:**

Eric Heitz. A Simpler and Exact Sampling Routine for the GGX Distribution of Visible Normals. [Research Report] Unity Technologies. 2017. <hal-01509746>

**HAL Id: hal-01509746**

**<https://hal.science/hal-01509746v1>**

Submitted on 18 Apr 2017

**HAL** is a multi-disciplinary open access archive for the deposit and dissemination of scientific research documents, whether they are published or not. The documents may come from teaching and research institutions in France or abroad, or from public or private research centers.

L'archive ouverte pluridisciplinaire **HAL**, est destinée au dépôt et à la diffusion de documents scientifiques de niveau recherche, publiés ou non, émanant des établissements d'enseignement et de recherche français ou étrangers, des laboratoires publics ou privés.



HAL Authorization

# A Simpler and Exact Sampling Routine for the GGX Distribution of Visible Normals

Eric Heitz

Unity Technologies

## Abstract

Heitz and d'Eon [HD14a] introduced importance sampling based on the distribution of visible normals (VNDF) and provided analytic solutions for Beckmann and GGX distributions [HD14b].

In this document, we propose an alternative sampling routine for the GGX VNDF that is simpler and exact. The solution provided by Heitz and d'Eon was approximate for GGX because it required a fitted curve: the derivations were made in slope space and the conditional inverse CDF of GGX does not have a closed form with this parameterization. This improved sampling routine uses the property that for unit roughness  $\alpha = 1$  the GGX distribution is a uniform hemisphere [HDCD15]. Sampling the GGX VNDF is thus equivalent to sampling the projected area of a hemisphere, which can be done exactly.




Figure 1: **Importance sampling the projected area of a hemisphere.** The half disks yield the silhouette of the hemisphere projected onto direction  $V$ . We sample its projected area by sampling points  $P$  on these half disks and projecting them on the hemisphere onto direction  $V$ .

```
vec3 sampleGGXVNDP(vec3 V_, float alpha_x, float alpha_y, float U1, float U2)
{
    // stretch view
    vec3 V = normalize(vec3(alpha_x * V_.x, alpha_y * V_.y, V_.z));

    // orthonormal basis
    vec3 T1 = (V.z < 0.9999) ? normalize(cross(V, vec3(0,0,1))) : vec3(1,0,0);
    vec3 T2 = cross(T1, V);

    // sample point with polar coordinates (r, phi)
    float a = 1.0 / (1.0 + V.z);
    float r = sqrt(U1);
    float phi = (U2 < a) ? U2/a * M_PI : M_PI + (U2-a)/(1.0-a) * M_PI;
    float P1 = r*cos(phi);
    float P2 = r*sin(phi)*(U2 < a) ? 1.0 : V.z;

    // compute normal
    vec3 N = P1*T1 + P2*T2 + sqrt(max(0.0, 1.0 - P1*P1 - P2*P2))*V;

    // unstretch
    N = normalize(vec3(alpha_x*N.x, alpha_y*N.y, max(0.0, N.z)));
    return N;
}
```

# 1 Algorithm

**Stretch configuration** As explained by Heitz and d'Eon [HD14a], we obtain the view vector in the configuration where  $\alpha_x = \alpha_y = 1$  by scaling its  $x$  and  $y$  components respectively  $\alpha_x$  and  $\alpha_y$  and renormalizing it.

**Orthonormal basis around the view vector** We construct an orthonormal basis  $(V, T_1, T_2)$  where  $T_1$  is in the tangent plane:

$$T_1 = \frac{V \times (001)}{\|V \times (001)\|}, \quad (1)$$

$$T_2 = T_1 \times V. \quad (2)$$

**Projected areas of the half disks** The projected area of the hemisphere is the sum of the projected areas of the two half disks show in Figure 1. The projected area of the half disk located in the tangent plane (in green) is proportional to  $Z \cdot V = z_v$  and the projected area of the other half disk (in blue) is proportional to  $T_2 \cdot V = 1$ .

**Sampling the half disks** We sample the projected area by generating samples on the half disks proportionally to their respective projected areas, i.e. with respective probabilities  $a = \frac{1}{1+z_v}$  and  $1 - a$ . For this, we use a polar parameterization of the disk  $(r, \phi)$  in which we scale the angle  $\phi$  in order to account for the difference of projected areas of its two halves (see Figure 2):

$$r = \sqrt{\mathcal{U}_1} \quad (3)$$

$$\phi = \begin{cases} \frac{\mathcal{U}_2}{a} \pi & \text{if } \mathcal{U}_2 < a, \\ \frac{\mathcal{U}_2 - a}{1 - a} \pi + \pi & \text{otherwise.} \end{cases} \quad (4)$$

If the sampled point is in the tangent half disk ( $\mathcal{U}_2 \geq a$ , in green) we obtain its projection in the view-orthogonal plane  $(T_1, T_2)$  by scaling its second component by  $z_v$ :

$$P = \begin{cases} r \cos(\phi) T_1 + r \sin(\phi) T_2 & \text{if } \mathcal{U}_2 < a, \\ r \cos(\phi) T_1 + r \sin(\phi) z_v T_2 & \text{otherwise.} \end{cases} \quad (5)$$




Figure 2: Sampling a point in the two half disks with different weights.

**Computing the normal** We obtain the normal  $N$  by projecting  $P$  on the sphere onto direction  $V$ :

$$N = P + \sqrt{1 - \|P\|^2} V. \quad (6)$$

**Unstretch the normal** Finally, to obtain the normal in the original configuration, we scale the  $x$  and  $y$  components of  $N$  by respectively  $\alpha_x$  and  $\alpha_y$  and we renormalize.

## 2 Remarks

**The view direction is located in the upper hemisphere** This sampling routine works under the assumption that if the view direction  $V$  is located in the upper hemisphere, i.e.  $z_v \geq 0$ . This is usually an assumption of the BRDF models. However, there are special cases where this sampling routine cannot be used as is, for instance the Smith multiple-scattering BRDF model [HHdD16] models incident rays that can be located in the lower hemisphere after their first bounce on the microsurface.

**Floating-point precision** In the proposed implementation, the function has two `max(0.0, ..)` instructions in order to avoid errors due to floating-point precision.

**Improved sampling routine for the Beckmann distribution** See also Wenzel Jakob’s improvement for the Beckmann distribution [Jak14].

## References

- [HD14a] Eric Heitz and Eugene D’Eon. Importance sampling microfacet-based BSDFs using the distribution of visible normals. In *Proc. Eurographics Symposium on Rendering*, pages 103–112, 2014. ([document](#)), 1
- [HD14b] Eric Heitz and Eugene D’Eon. Importance sampling microfacet-based BSDFs using the distribution of visible normals, supplemental material. <https://hal.inria.fr/hal-00996995v2/file/supplemental1.pdf>, 2014. ([document](#))
- [HDCD15] Eric Heitz, Jonathan Dupuy, Cyril Crassin, and Carsten Dachsbacher. The SGX microflake distribution. *ACM Trans. Graph.*, 34(4):48:1–48:11, 2015. ([document](#))
- [HHdD16] Eric Heitz, Johannes Hanika, Eugene d’Eon, and Carsten Dachsbacher. Multiple-scattering microfacet bsdfs with the smith model. *ACM Trans. Graph.*, 35(4):58:1–58:14, 2016. 2
- [Jak14] Wenzel Jakob. An improved visible normal sampling routine for the beckmann distribution. Technical report, ETH Zurich, August 2014. 2